

De MUSIC à SOUND

C'est la mère Michel (2)

par André C.

Syntaxe et usages

MUSIC et SOUND sont les deux commandes qui génèrent des sons. Les syntaxes sont comparables :

MUSIC, canal, octave, note, volume

SOUND, canal, hauteur, volume.

Dans les deux cas, il faut indiquer sur quel canal le son doit être généré et le volume sonore désiré. En pratique, MUSIC est plus simple à utiliser car il suffit d'indiquer l'octave et la note voulue. Mais ne peut générer que des fréquences prédéfinies. Par contre, la commande SOUND présente un gros avantage : elle peut générer n'importe quelle fréquence, pourvu que celle-ci soit audible par l'oreille humaine. Mais pour produire de la musique, il faut connaître le fréquence des notes à jouer.

Hauteur des notes

Tout repose sur la fréquence du LA de l'octave 3 dont la valeur a été définie internationalement à 440 Hz. La fréquence de toutes les notes en découle. La fréquence de base de chaque octave est un multiple de 440 Hz. Ainsi le LA de l'octave 4 a une fréquence de 880 Hz, tandis que celle du LA de l'octave 2 est de 220 Hz, etc. Chaque octave est divisé en 12 intervalles et pour passer d'un note à la suivante, il faut multiplier sa fréquence par la racine douzième de 2. La Figure 1 ci-dessous montre les valeurs que l'on trouve sur Wikipédia à l'adresse suivante :

https://fr.wikipedia.org/wiki/Note_de_musique (page très intéressante par ailleurs).

Étant d'un naturel paresseux (une qualité propre à tous les génies), je me suis dit qu'il serait fastidieux de taper une à une les fréquences de toutes les notes, d'autant plus que la commande SOUND n'accepte pas directement la fréquence, mais un paramètre $P=62480/\text{fréquence}$. Comme si la routine SOUND n'avait pas pu faire ce calcul elle-même !

Il m'a semblé plus astucieux de calculer la fréquence de toutes les notes ainsi que les paramètres correspondants et de placer automatiquement ces valeurs dans un tableau, plutôt que de jouer du clavier !

J'ai eu un peu de mal à trouver une valeur suffisamment précise pour la racine douzième de 2.

En effet, les décimales de cette racine s'étendent à l'infini. La meilleure valeur que j'ai obtenue en calculant avec l'Atmos pour la racine est $R=1.05946309419999$. Mais cette précision est insuffisante, car même si l'erreur est faible, quand elle est répétée de manière récurrente, elle devient sensible pour les notes qui sont les plus éloignées du LA de l'octave 3. Les fréquences calculées et par suite les paramètres correspondants s'écartent trop des valeurs trouvées sur Wikipédia (Figure 1).

La valeur que l'on peut calculer dépend étroitement des performances du microprocesseur utilisé pour le

Figure 1

Fréquences des notes (en hertz) dans la gamme tempérée

Note/octave	0	1	2	3	4	5	6	7
do ou s#	32,70	65,41	130,81	261,63	523,25	1046,50	2093,00	4186,01
do# ou réb	34,65	69,30	138,59	277,18	554,37	1108,73	2217,46	4434,92
ré	36,71	73,42	146,83	293,66	587,33	1174,66	2349,32	4698,64
ré# ou mi♭	38,89	77,78	155,56	311,13	622,25	1244,51	2489,02	4978,03
mi ou fa♭	41,20	82,41	164,81	329,63	659,26	1318,51	2637,02	5274,04
fa ou mi#	43,65	87,31	174,61	349,23	698,46	1396,91	2793,83	5587,65
fa# ou so♭	46,25	92,50	185,00	369,99	739,99	1479,98	2959,96	5919,91
so♭	49,00	98,00	196,00	392,00	783,99	1567,98	3135,96	6271,93
so# ou la♭	51,91	103,83	207,65	415,30	830,61	1661,22	3322,44	6644,88
la	55,00	110,00	220,00	440,00	880,00	1760,00	3520,00	7040,00
la# ou si♭	58,27	116,54	233,08	466,16	932,33	1864,66	3729,31	7458,62
si ou do♭	61,74	123,47	246,94	493,88	987,77	1975,53	3951,07	7902,13

Figure 3

O	N	F	P
2	DO	130,81	477,63
2	DO#	138,59	450,82
2	RE	146,83	425,52
2	RE#	155,56	401,64
2	MI	164,81	379,09
2	FA	174,61	357,82
2	FA#	185,00	337,73
2	SOL	196,00	318,78
2	SOL#	207,65	300,89
2	LA	220,00	284,00
2	LA#	233,08	268,06
2	SI	246,94	253,01
3	DO	261,63	238,81
3	DO#	277,18	225,41
3	RE	293,67	212,76
3	RE#	311,13	200,82
3	MI	329,63	189,55
3	FA	349,23	178,91
3	FA#	369,99	168,87
3	SOL	392,00	159,39
3	SOL#	415,31	150,44
3	LA	440,00	142,00
3	LA#	466,16	134,03
3	SI	493,88	126,51
4	DO	523,25	119,41
4	DO#	554,37	112,71
4	RE	587,33	106,38
4	RE#	622,25	100,41
4	MI	659,26	94,77
4	FA	698,46	89,45
4	FA#	739,99	84,43
4	SOL	783,99	79,69
4	SOL#	830,61	75,22
4	LA	880,00	71,00
4	LA#	932,33	67,01
4	SI	987,77	63,25

```

100 F=123.471:' SI de l'octave 2
110 R=1.05946309435534:' Racine 12e de 2
120 H=62480:' Facteur de conversion
130 DIM F(36):' Tableau Frequences
140 DIM P(36):' Tableau Parametres
150 J=1
160 FOR I=1 TO 36
170 J=J*R
180 F(I)=F*J
190 P(I)=H/F(I)
200 NEXT
    
```

Figure 2

calcul. J'ai dû faire appel à mon PC et je suis arrivé à R=1,05946309435534. Les 5 dernières décimales sont notablement différentes de celles calculées avec l'Oric.

Cette fois-ci, en utilisant R=1,05946309435534, j'ai obtenu exactement les mêmes résultats avec mon Atmos et avec mon PC et ces résultats sont parfaitement conformes aux données de Wikipédia.

Pas si génial que ça le gros paresseux : à calculer tout ça, j'ai passé dix fois plus de temps que si j'avais tapé les 36 valeurs requises dans mon programme Basic ! Disons que je me suis fait plaisir...

Voici le programme utilisé (Figure 2, ci-dessus) et le tableau des résultats obtenus (Figure 3, ci-contre).

Les fréquences obtenues sont identiques à celles données par Wikipédia. Il ne reste plus qu'à adapter le programme donné dans l'article précédent ("Transcrivez une partition musicale en programme Oric, avec un exemple : C'est la mère Michel") en remplaçant les commandes MUSIC par des commandes SOUND. Il "suffit" de remplacer les paires de valeurs octave/note par un paramètre P adapté à la fréquence des notes à produire. Ce programme ne fournit que les valeurs pour les octaves 2, 3 et 4 mais peut être adapté si besoin. Le tableau des Paramètres P(I) généré par le programme Basic peut être exploité avec la commande SOUND.

Rappel : La partition de "C'est la Mère Michel"

Voir la Figure 4, ci-dessous.

Le programme correspondant avec SOUND

Voir la Figure 6, page suivante.

Figure 4

Les notes de 1 à 12 pour les octaves n°3 et n°4 : Paramètres pour SOUND

do	ré	mi	fa	sol	la	si	do	ré	mi	fa	sol	la	si
238.81	212.76	189.55	178.91	159.39	142.00	126.51	119.41	106.38	94.77	89.45	79.69	71.00	63.25
do#	ré#		fa#	sol#	la#		do#	ré#		fa#	sol#	la#	
225.41	200.82		168.87	150.44	134.03		112.71	100.41		84.43	75.22	67.01	

Figure 5

```

80 ' C'est la mere Michel
90 ' Calcul des parametres pour Sound
100 F=123.471:' Frequence de la note SI de l'octave 2
110 R=1.05946309435534:' Racine 12e de 2
120 H=62480:' Facteur de conversion
140 DIM P(36):' Tableau Parametres
150 J=1
160 FOR I=1 TO 36
170 J=J*R
190 P(I)=H/(F*J)
220 NEXT
1010 V=10:' Volume, Octave, Note, Duree, Silence
1020 FOR J=1 TO 73
1030 PLAY 1,0,0,0
1040 READ O,N,D,S:I=((O-2)*12)+N
1050 SOUND 1,P(I),V:WAIT D:PLAY 0,0,0,0:WAIT S
1060 NEXT
1070 CALL#C4B7:' Pseudo END, evite Ready et parasite sonore
2010 DATA 3,10,30,0, 3,10,10,2, 3,10,20,0, 3,10,20,2: 'C'est la mer' Mi
2020 DATA 3,10,20,0, 3, 5,20,2, 3,10,20,0, 3,12,20,2: 'chel qui a per
2030 DATA 4, 2,40,2, 3,12,40,2: 'du son
2040 DATA 3,10,60,2, 3,10,20,2: 'chat Qui
2050 DATA 3, 8,30,0, 3, 6,10,2, 3, 8,20,0, 3,10,20,2: 'cri par la fe
2060 DATA 3,12,20,0, 4, 2,20,2, 3,12,20,0, 3,10,20,2: 'netre A qui le
2070 DATA 3, 8,40,2, 3, 6,40,2: 'lui ren
2080 DATA 3, 5,60,2, 3, 5,20,2: 'dra C'est
2090 DATA 3,12,20,0, 3,12,20,2, 3,12,20,0, 4, 2,20,2: 'le pere Lus tu
2100 DATA 4, 3,60,2, 4, 3,20,2: 'cru qui
2110 DATA 4, 2,20,0, 3,10,20,2, 4, 5,20,0, 4, 2,20,2: 'lui a re pon
2120 DATA 3,12,60,2, 3,12,20,2: 'du Al
2130 DATA 4, 3,20,0, 4, 3,20,2, 4, 3,20,0, 4, 3,20,2: 'lez la mere Mi
2140 DATA 4, 2,20,0, 4, 5,20,2, 4, 2,20,0, 3,10,20,2: 'chel Vot chat n'est
2150 DATA 3,12,40,2, 4, 5,40,2: 'pas per
2160 DATA 3,10,20,2, 3,10,20,2, 3,10,20,0, 3, 8,20,2: 'du Sur l'air du
2170 DATA 3, 6,40,2, 3, 6,30,0, 3, 6,10,2: 'tra la la
2180 DATA 3, 6,20,0, 3, 8,20,2, 3,10,20,0, 3,12,20,2: 'la Sur l'air du
2190 DATA 3, 5,40,2, 3, 5,30,0, 3, 5,10,2: 'tra la la
2200 DATA 3, 5,60,2, 3, 5,20,2: 'la Sur
2210 DATA 3,10,40,2, 4, 2,40,2: 'l'air du
2220 DATA 4, 5,20,0, 4, 3,20,2, 4, 2,20,0, 3,12,20,2: 'tra de ri de
2230 DATA 3,10,20,0, 4, 5,20,2, 4, 5,20,0, 4, 5,20,2: 'ra Et tra la
2240 DATA 3,10,80,2: 'la

```

Si on avait à taper le programme d'une nouvelle partition, pour toutes les commandes SOUND, il faudrait placer dans les DATA, soit les différentes valeurs du paramètre P, soit les différentes valeurs de l'indice I du tableau P(I). La [Figure 5, page précédente](#), vous montre ce qui vous attend au cas où vous voudriez taper les paramètres eux-mêmes (valeurs complexes à entrer sans erreurs) ! Il est plus simple d'utiliser le tableau des Paramètres P(I) calculé précédemment et d'indiquer l'indice du tableau correspondant à chaque note. Mais, dans le cas présent, comme le gros travail de déchiffrement de la partition ci-dessus a déjà été fait et que je reste malgré tout un génie (oups, je voulais dire un paresseux !), j'ai trouvé plus simple de calculer l'indice I à partir de chaque couple octave/note, plutôt que de modifier tous les DATA. Voilà ce que ça donne ([Figure 6, ci dessus](#)).

Ce programme fait un peu bricolé, mais c'est intentionnellement que j'ai conservé au maximum la version précédente qui utilisait la commande MUSIC. J'ai ajouté par devant la conversion des paires

"Octave/Note" de la commande MUSIC en indices du tableau P pour la commande SOUND.

Vous trouverez ce listing ainsi que la disquette du programme dans le fichier MereMichel2.zip qui accompagne ce CEO-mag. Vous constaterez qu'il n'y a aucune différence entre les mélodies produites avec MUSIC ou avec SOUND.

Conclusion / Perspectives

Tant que nous y sommes, il faudrait améliorer cette musique en ajoutant des harmoniques, ce que SOUND permet. C'est nettement moins facile, mais cela offre plus de possibilité. Quant à transcrire ce programme en langage machine, cela ne présente aucun intérêt, sauf pour la jouer en bi-tâche, par exemple pour un jeu avec une musique de fond. Pour cela voyez l'article de Dominique Pessan : "Interruptions musicales" (CEO-mag n°99 de juillet 1998, pages 20 à 22).

Enfin je vous invite à transcrire d'autres airs afin d'enrichir le répertoire musical de l'Oric...

Bon amusement.  